



Speaker Angles II Calculating The Ideal Speaker Location

REVIEW

• Did you save July/August's issue like I asked? Good. Well, here is the second program. Let's review the four programs briefly.

There are usually a limited number of places to mount the speaker. The first program will tell the ideal required coverage angles for an actual mounting location.

After the ideal angles are calculated, a speaker should be selected to closely match the ideal. The second program now calculates the ideal location for the actual coverage angles.

Enter practical compromise. Adjust between the two previous locations for a usable location, and the third program will calculate the actual loss for an actual coverage angle and an actual location. This allows for testing the variation of the SPL

coverage without climbing a scaffold or buying the wrong speaker.

If an existing system is being analyzed, the third program will tell the variation of coverage. If it is not uniform enough, the second program will indicate the proper mounting location. If the proper mounting location is ten feet above the ceiling (or any other unsuitable location), then the process of selecting a new speaker starts with the first program as described above.

The fourth program calculates the direct path from the speaker to the microphone, and the primary reflection path off the back wall, to aid in locating the acoustic padding.

PROGRAM

The "P.A. Position" program is very similar to the "P.A. Angle" program. They operate the same way,

and have nearly identical screens. Most of the program code is the same. The primary difference is that the speaker coverage angle and the distance from the first to last listener is entered instead of the speaker location. The calculations are more complicated also. The core of the program (lines 100 through 9020) is identical, and can be copied from the first program to save time typing. *Figure 1* is a sample screen to test your program.

ANGLE LIMITS

This program more clearly reveals a severe limitation on allowable coverage angles as indicated in the last issue. You will also find that only one -12dB angle is correct for a given -6dB angle. I can hear you now, "But they don't make speakers like that!" Tell me about it! On the other hand, tell the people who make the speakers. As I said before, this is not a new concept or science.

OOPS!

If you look at *Figure 3* in *db's* July/August issue, you will notice "-6 dB BACK WALL:" was followed by 25 feet 12 inches, which should have calculated as 26 feet. This rounding error is easily corrected by changing the last routine of the first program (lines 51500 through 51530) to match the second program (replace with lines 51500 through 51560).

In the first program, line 50500 should be: "50500 REM VERT DIST FROM FLOOR TO AVERAGE LISTENING HEIGHT".

We had a couple of minor typos (probably because I was pushing deadline) last time. The equation on

Figure 1. A sample screen.

P.A. SPEAKER POSITION CALCULATIONS

H. 1ST TO LAST LIST: 100 ft 0 in.
H. LAST LIST TO WALL: 10 ft 0 in.

V. FLOOR TO AV LIST: 4 ft 6 in.
SPKR -6 dB COVR ANGL: 20 ft
ON AXIS SPKR TO REF: 4 ft 0 in.

SPEAKER TO:	Vertical height Feet In.	Horizontal distance Feet In.	Throw distance Feet In.	dB loss Decibels	Vertical to throw Degrees
LAST LISTENER:	21 1	123 1	124 10	-29.89	80.29
-6 dB LISTENER:	21 1	56 9	62 5	-29.86	70.29
-12 dB LISTENER:	21 1	23 1	31 2	-29.84	47.59
-6 dB BACK WALL:	26 3	133 1	133 1	-36.44	90.29
-12 dB BACK WALL:	82 0	133 1	144 6	-43.16	112.99

-6 dB COVERAGE ANGLE: 20.00°
-12 dB COVERAGE ANGLE: 65.40°
SPEAKER TILT FROM VERT: 9.71°

VERT FLR TO CENT OF SPKR: 25 ft 7 in.
AVG LIST TO CENT OF SPKR: 21 ft 1 in.

/CR/TO START AGAIN:

page 15 is really an equation and an example.

The “ $-12dB=20\text{Log}(4\text{ft./16ft.})$ ” is the example. Page 16, *Figure 2*, should be “D=Distance” (as if you couldn’t guess) and D6 goes from L3 to L1.

TRICKY TYPING

In this marvelous age of technology, computers and typesetting machines still do not agree well with each other. If you are not much of a programmer, you probably had some difficulty typing the first program in.

Here are a couple of tips. Many of the program lines were longer than the column width, so they were split on two or three lines.

A new line is indented one space and always begins with a line number. If the line has been split, you need to put it all back on one line or the computer will get nasty. You may also have some difficulty knowing how many blank spaces to use in some places. My best advice is trial and error. If the screen doesn’t look right, check lines 11000 though 11125 for the background text and

the prompts. Check 50000 through 50390 for the display of the results. P\$=” is as important as a PRINT line.

STAY TUNED

As I said last time, save this issue and make sure your subscription is up to date. Next program, next issue.

If you have any questions or comments, please contact me through **db Magazine**.

The Basic Program

```

10 REM SPEAKER POSITION
CALCULATIONS FOR P.A. SYSTEM
20 REM "PAPOSITN.BAS",A
30 REM V1.1
40 REM 07-12-90
50 REM DCR

100 REM ***** INITIALIZE
110 REM **** SYSTEM FUNCTIONS
120 ON ERROR GOTO 8000
130 CLEAR
140 KEY OFF

800 REM **** SET SYSTEM
VARIABLES
810 P3$=STRING$(80," ")
820 P4$=STRING$(80,"-")
830 P5$=STRING$(80,"=")
840 P6$=CHR$(254)

900 REM **** SET PROGRAM
VARIABLES
910 GOSUB 10000

1000 REM ***** DISPLAY SCREEN

1010 REM **** INITIALIZE
1020 CLS
1030 COLOR C0,C1,C
1040 FOR Y=1 TO 25
1050 LOCATE Y,1:PRINT P3$;
1060 NEXT Y
1100 REM **** HEADING
1110 REM *** FRAME
1120 LOCATE 3,1:PRINT P5$;
1130 LOCATE 22,1:PRINT P5$;

1140 IF G0>0 THEN LOCATE
G1,G0:PRINT LEFT$(P4$,G2);
1150 IF G3>0 THEN LOCATE
G4,G3:PRINT LEFT$(P4$,G5);
1200 REM *** SYSTEM FUNCTIONS
1210 GOSUB 6000
1220 REM *** TITLE
1230 COLOR C0,C1,C
1240 LOCATE 2,1:PRINT G$;

1300 REM **** LINES
1310 REM *** INITIALIZE
1320 RESTORE
1330 J1=0
1340 ON ERROR GOTO 1900
1350 REM *** GET DATA
1360 READ
F$,F0$,F1$,F2$,F,F0,F1,F2,F3,F4,F3$
1370 J1=J1+1
1400 REM *** SET NUMBER
1410 P1$=""
1420 IF F0$="N" THEN GOTO 1500
1430 P1$=STR$(J1)
1440 FOR J3=1 TO LEN(P1$):IF
LEFT$(P1$,1)=" " THEN LET
P1$=RIGHT$(P1$,2):NEXT J3
1450 IF F0$="0" THEN GOTO 1490
1460 IF LEN(P1$)=1 THEN LET
P1$="0"+P1$
1470 IF F0$="2" THEN IF LEN(P1$)=2
THEN LET P1$="0"+P1$
1490 P1$=P1$+" "
1500 REM *** DISPLAY
1510 LOCATE F0,F:PRINT P1$+F$;
1600 REM *** REPEAT
1610 GOTO 1350

1900 REM **** END OF DISPLAY
1910 RESUME 1920

1920 ON ERROR GOTO 8000

2000 REM ***** INPUT DATA

2010 REM **** INITIALIZE
2020 RESTORE

2050 REM **** START LOOP
2060 FOR J=1 TO J1

2070 REM **** GET PARAMETERS
2080 READ
F$,F0$,F1$,F2$,F,F0,F1,F2,F3,F4,F3$

2100 REM **** PROMPTS
2110 COLOR C0,C1,C
2120 LOCATE 23,1:PRINT P3$;
2130 LOCATE 24,1:PRINT P3$;
2140 COLOR C2,C3,C
2150 LOCATE: 23,1:PRINT F3$;

2200 REM **** GET INPUT
2210 GOSUB 7000

2300 REM **** VALIDATE
2310 IF LEN(D$)><1 THEN GOTO
2350
2320 IF J=1 THEN IF
INSTR("QqEeXtT",D$)>0 THEN GOTO
9000
2330 IF ASC(D$)=27 THEN GOTO
8100
2340 IF D$="!" THEN GOTO 6100
2345 IF D$="*" THEN GOTO 100
2350 FLAG$=""
2360 GOSUB 20000

```

```

30090 IF J=9 THEN LET A9$=D$
39999 RETURN

40000 REM ***** CALCULATIONS -
USER SUBROUTINE

40010 REM ***** CONVERT TO
NUMBERS AND INCHES
40020 REM *** FIRST TO LAST
LISTENER
40030 D6#=(12*VAL(A1$))+VAL(A2$)
40050 REM *** LAST LISTENER TO
WALL
40060 D4#=(12*VAL(A3$))+VAL(A4$)
40070 REM *** FLOOR TO AVERAGE
LISTENING HEIGHT
40080 H6#=(12*VAL(A5$))+VAL(A6$)
40090 REM *** DISTANCE FROM
SPEAKER FOR db SPL REFERENCE
MEASUREMENT
40100 R#=(12*VAL(A8$))+VAL(A9$)

40110 REM *** SPEAKER'S RATED
-6dB COVERAGE ANGLE
40120 A4#=VAL(A7$)

40130 REM *** CONVERSION
FACTORS
40140 REM ** RADIANS TO DEGREES
40150 RD#=180/3.1415927
40160 REM ** DEGREES TO RADIANS
40170 DR#=3.1415927/180

40200 REM ***** SOLVE FOR ANGLES
FROM HORIZONTAL TO THROW
40210 REM *** A1 - VERTICAL TO ON
AXIS THROW ANGLE
40220
A1#=(RD#*ATN((2-COS((A4#/2)*DR#)
)/(SIN((A4#/2)*DR#))))
40230 REM *** A2 - VERTICAL TO
-6dB AXIS ANGLE
40240 A2#=(A1#-(A4#/2))

40250 REM *** A3 - VERTICAL TO
-12dB AXIS ANGLE
40260 REM
A3#=ARCCOSINE(4*COS(A1#*DR#))

40262
GOAL#=INT(4*COS(A1#*DR#)*10000
00)
40264 IF GOAL#>=1000000 THEN
LET A3#=A3#/0
40266 TOPANGLE#=90
40268 BOTTOMANGLE#=.01
40270
TESTANGLE#=((TOPANGLE#-BOTTO
MANGLE#)/2)+BOTTOMANGLE#
40272
TEST#=INT(COS(TESTANGLE#*DR#)
*1000000)

40274 IF TEST#=GOAL# THEN
GOTO 40282
40276 IF TEST#<GOAL# THEN LET
TOPANGLE#=TESTANGLE#
40278 IF TEST#>GOAL# THEN LET
BOTTOMANGLE#=TESTANGLE#
40280 GOTO 40270
40282 A3#=TESTANGLE#

40300 REM ***** D1 - HORIZONTAL
SPEAKER TO LAST LISTENER
40310
D1#=D6#/(1-(TAN(A3#*DR#)/TAN(A1#
*DR#)))

40320 REM ***** H1,H2,H3 - SPEAKER
HEIGHT ABOVE AUDIENCE
40330 REM *** H1
40340 H1#=D1#/TAN(A1#*DR#)
40350 H2#=H1#
40360 H3#=H1#
40370 H7#=H1#+H6#

40400 REM ***** REMAINING
HORIZONTAL DISTANCES
40410 REM *** D2 - HORIZONTAL
SPEAKER TO -6dB AXIS
40420 D2#=H1#*TAN(A2#*DR#)
40430 REM *** D3 - HORIZONTAL
SPEAKER TO -12dB AXIS
40440 D3#=H1#*TAN(A3#*DR#)
40450 REM *** SPEAKER TO BACK
WALL
40460 D5#=D1#+D4#

40500 REM ***** CALCULATE
SPEAKER AXIS THROW DISTANCES
40510 REM *** 0 db ON AXIS THROW
DISTANCE
40520 T1#=(H1#/COS(A1#*DR#))
40530 REM *** -6 db AXIS THROW
DISTANCE
40540 T2#=(H1#/COS(A2#*DR#))
40550 REM *** -12 db AXIS THROW
DISTANCE
40560 T3#=(H1#/COS(A3#*DR#))

40670 REM ***** SPEAKER TILT FROM
VERTICAL
40680 A6#=90-A1#

40700 REM ***** CALCULATE
SPEAKER COVERAGE ANGLES
40730 REM *** -12 db ANGLE
40740 A5#=(A1#-A3#)*2

40800 REM ***** CALCULATE BACK
WALL REFLECTION

40810 REM *** -6 db
40820 REM ** ANGLE FROM
HORIZONTAL IN DEGREES
40830 A9#=A4#+A2#-90
40840 REM ** THROW DISTANCE
40850
T4#=D5#/COS(ABS(A9#*DR#))
40860 REM ** BACK WALL HEIGHT
40870
H4#=H7#+((ABS(A9#)/A9#)*((TAN(A
BS(A9#*DR#)))*D5#))
40880 REM ** ANGLE FROM
VERTICAL
40890 A7#=A1#+A1#-A2#

40900 REM *** -12 db
40910 REM ** ANGLE FROM
HORIZONTAL IN DEGREES
40920 A9#=A5#+A3#-90
40930 REM ** THROW DISTANCE
40940
T5#=D5#/COS(ABS(A9#*DR#))
40950 REM ** BACK WALL HEIGHT
40960
H5#=H7#+((ABS(A9#)/A9#)*((TAN(A
BS(A9#*DR#)))*D5#))
40970 REM ** ANGLE FROM
VERTICAL
40980 A8#=A1#+A1#-A3#

41000 REM ***** CALCULATE
DISTANCE SPL LOSS FOR AXIS
41010 REM *** 0 db ON AXIS TO
AUDIENCE
41020
L1#=20*(LOG(R#/T1#)/LOG(10))
41030 REM *** -6 db AXIS TO
AUDIENCE
41040
L2#=20*(LOG(R#/T2#)/LOG(10))-6
41050 REM *** -12 db AXIS TO
AUDIENCE
41060
L3#=20*(LOG(R#/T3#)/LOG(10))-12
41100 REM *** -6 db AXIS TO BACK
WALL
41110
L4#=20*(LOG(R#/T4#)/LOG(10))-6
41120 REM *** -12 db AXIS TO BACK
WALL
41130
L5#=20*(LOG(R#/T5#)/LOG(10))-12

49999 RETURN

50000 REM ***** DISPLAY RESULTS -
USER SUBROUTINE
50010 REM ***** TEXT FORMAT
50020 REM *** DISPLAY FRAME
50030 LOCATE 8,1:PRINT "
VERTICAL HORIZONTAL THROW
VERTICAL"

```

```

7970 LOCATE F2,F1+LEN(D$):PRINT
P2$;
7980 RETURN

8000 REM ***** ERRORS
8010 RESUME 8020
8020 COLOR C14,C15,C
8030 LOCATE 23,1:PRINT P3$;
8040 LOCATE 24,1:PRINT P3$;
8050 COLOR C14,C15,C
8040 LOCATE 23,1:PRINT "ERROR AT
LINE";ERL;
8050 LOCATE 24,1:PRINT E$;
8060 INPUT "",X$
8070 GOTO 1000

8100 REM ***** STOP
8110 ON ERROR GOTO 0
8120 COLOR 15,0,0
8130 STOP

9000 REM ***** EXIT
9010 CLS
9020 SYSTEM

10000 REM ***** PROGRAM
VARIABLES

10010 REM **** PROGRAM TITLE
10020 LET G$="P.A. SPEAKER
POSITION CALCULATIONS"

10030 REM **** ERROR MESSAGE
10040 LET E$="CONFIGURATION IS
NOT POSSIBLE. ANY KEY TO
RESTART: "

10050 REM **** BELL AT AFTER
MASK DISPLAY
10060 LET BELL1$="N"

10070 REM **** BELL AT FIELD FULL
PROMPT
10080 LET BELL2$="N"

10100 REM **** DIVIDING LINES
X,Y,LEN
10110 LET G0=1:LET G1=7:LET
G2=80
10120 LET G3=0:LET G4=0:LET
G5=80
10200 REM **** COLORS
10210 REM *** BORDER
10220 C = 0

10230 REM - ALL OTHERS TEXT &
BACKGROUND
10240 REM *** INITIALIZE AND
BACKGROUND TEXT
10250 C0 = 7 : C1 = 0
10260 REM *** PROMPTS
10270 C2 = 15 : C3 = 0
10280 REM *** MASK
10290 C4 = 15 : C5 = 0
10300 REM *** CURRENT CURSOR
10310 C6 = 31 : C7 = 0
10320 REM *** CURRENT INPUT
10330 C8 = 15 : C9 = 0
10340 REM *** FOREGROUND TEXT
(ACCEPTED INPUT)
10350 C10 = 15 : C11 = 0
10360 REM *** FOREGROUND TEXT
(OUTPUT DISPLAY)
10370 C12 = 15 : C13 = 0
10380 REM *** ERROR TEXT
10390 C14 = 15 : C15 = 0

10999 RETURN

11000 REM ***** DATA
11005 REM FIELD DESC,AUTO
NO.,DEFAULT,MASK CHR,X,Y,IN
X,Y,MIN,MAX,PROMPT
11006 REM
F$,F0$,F1$,F2$,F,F0,F1,F2,F3,F4,F3$
11007 REM AUTO NO.: N= OMIT
NUMBER, 0= OMIT LEADING ZERO,
1= 2 DIGIT NO., 2= 3 DIGIT NO.
11008 REM MASK CHARACTER (F2$)
= TO "&" WILL DISPLAY A BOX -
CHR$(254)

11010 DATA "H 1ST TO LAST LISTN:
ft"
11011 DATA "N","0","&",1,4,22,4,0,4
11012 DATA "HORIZ. DISTANCE
FROM THE FIRST TO LAST LISTENER
(FEET + INCHES OR INCHES ONLY)"

11020 DATA "in"
11021 DATA "N","0","&",37,4,30,4,0,6
11022 DATA "HORIZ. DISTANCE
FROM THE FIRST TO LAST LISTENER
(INCHES ADDED TO FEET)"

11030 DATA "H LAST LIST TO WALL:
ft"
11031 DATA "N","0","&",1,5,22,5,0,4
11032 DATA "HORIZ. DISTANCE
FROM LAST LISTENER TO BACK
WALL"
11040 DATA "in"
11041 DATA "N","0","&",37,5,30,5,0,6

11043 DATA "HORIZ. DISTANCE
FROM LAST LISTENER TO BACK
WALL "

11050 DATA "V FLOOR TO AV LIST :
ft"
11051 DATA "N","0","&",42,4,63,4,0,4
11052 DATA "VERT. DISTANCE FROM
FLOOR TO AVG. LISTENING HEIGHT"

11060 DATA "in"
11061 DATA "N","0","&",76,4,71,4,0,4
11062 DATA "VERT. DISTANCE FROM
FLOOR TO AVG. LISTENING HEIGHT"

11070 DATA "SPKR -6dB COVR
ANGL:"
11071 DATA "N","0","&",42,5,63,5,0,6
11072 DATA "ENTER THE SPEAKER'S
RATED -6 dB COVERAGE ANGLE
(LESS THAN 31.05 DEG.)"

11080 DATA "ON AXIS SPKR TO REF:
ft"
11081 DATA "N","0","&",42,6,63,6,0,4
11082 DATA "DISTANCE FROM
SPEAKER TO ON AXIS db SPL REF.
MEASUREMENT"

11090 DATA "in"
11091 DATA "N","0","&",76,6,71,6,0,4
11092 DATA "DISTANCE FROM
SPEAKER TO ON AXIS db SPL REF.
MEASUREMENT"

20000 REM ***** VALIDATIONS -
USER SUBROUTINE
20010 IF D$="" THEN LET D$=F1$
20020 FOR J3=1 TO LEN(D$)
20030 IF
INSTR("0123456789.-+ ",MID$(D$,J3,1))=
0 THEN LET FLAG$="REENTER"
20040 NEXT J3
29999 RETURN

30000 REM ***** SLOT DATA - USER
SUBROUTINE
30010 IF J=1 THEN LET A1$=D$
30020 IF J=2 THEN LET A2$=D$
30030 IF J=3 THEN LET A3$=D$
30040 IF J=4 THEN LET A4$=D$
30050 IF J=5 THEN LET A5$=D$
30060 IF J=6 THEN LET A6$=D$
30070 IF J=7 THEN LET A7$=D$
30080 IF J=8 THEN LET A8$=D$

```



```

2370 IF FLAG$="REENTER" THEN
GOTO 2100
2380 IF FLAG$="START OVER" THEN
GOTO 100
2390 IF FLAG$="ERROR" THEN
GOTO 8000

2400 REM **** REDISPLAY
2410 COLOR C10,C11,C
2420 LOCATE F2,F1:PRINT D$;
2430 COLOR C0,C1,C
2440 PRINT
LEFT$(P3$,F4-LEN(D$)+1);

2500 REM **** SLOT DATA
2510 GOSUB 30000

2600 REM **** END OF LOOP
2610 NEXT J

2700 REM ***** CALCULATIONS
2710 GOSUB 40000

2800 REM ***** DISPLAY RESULTS
2810 COLOR C12,C13,C
2820 GOSUB 50000

3000 REM ***** END OF SCREEN

3010 REM **** PROMPT
3020 F$="/CR/ TO START AGAIN:"
3030 LET F0$="0":F2$="&"
3040
F=1:F0=23:F1=22:F2=23:F3=0:F4=1
3050 COLOR C0,C1,C
3060 LOCATE 23,1:PRINT P3$;
3070 LOCATE 24,1:PRINT P3$;
3080 COLOR C2,C3,C
3090 LOCATE F0,F:PRINT F$;
3100 GOSUB 7000
3110 GOTO 1000

6000 REM ***** DATE & TIME
SUBROUTINE
6010 COLOR C0,C1,C
6020 LOCATE 1,70:PRINT DATES;
6030 LOCATE 2,70:PRINT TIME$;
6040 LET PREVT$=TIME$
6050 RETURN

6100 REM ***** BACK-UP ONE FIELD
ROUTINE
6110 REM *** CLEAR CURRENT
FIELD
6120 COLOR C0,C1,C

6130 LOCATE F2,F1:PRINT
LEFT$(P3$,F4);
6140 IF J=1 THEN GOTO 100
6200 REM *** RESET FIELD
6210 RESTORE
6220 J2=J-1
6230 FOR J3=1 TO J2
6240 READ
F$,F0$,F1$,F2$,F,F0,F1,F2,F3,F4,F3$
6250 NEXT J3
6260 J=J3-1
6270 GOTO 2100

7000 REM ***** STANDARD
KEYBOARD INPUT SUBROUTINE

7010 REM **** MASK
7020 IF F2$="&" THEN LET F2$=P6$
7030 IF LEN(F2$)>1 THEN LET
P$=F2$:GOTO 7060
7040 IF F2$="" THEN LET
P$="":GOTO 7060
7050 LET P$=STRING$(F4,F2$)
7060 P$=P$+" "
7070 COLOR C4,C5,C
7080 LOCATE F2,F1:PRINT P$;
7090 IF BELL1$="Y" THEN PRINT
CHR$(7);
7095 REM --- SET BELL PARAMS &
GOSUB

7100 REM **** CLEAR INPUT
VARIABLE
7110 D$=""

7200 REM **** CHECK FOR FIELD
FULL
7210 IF LEN(D$)><F4 THEN GOTO
7300
7220 COLOR C2,C3,C
7230 LOCATE 24,1:PRINT "THIS FIELD
IS FULL. /CR/ OR BACKSPACE.";
7240 IF BELL2$="Y" THEN PRINT
CHR$(7);
7245 REM - SET BELL PARAMS &
GOSUB

7300 REM **** INPUT
7310 LOCATE F2,F1
7320 GOSUB 7900
7330 D1$=INKEY$
7340 IF TIME$><PREVT$ THEN
GOSUB 6000
7350 IF D1$="" THEN GOTO 7330
7360 GOSUB 7900

7400 REM **** /CR/ CHECK
7410 IF ASC(D1$)<>13 THEN GOTO
7600
7420 IF F3=0 THEN GOTO 7800

7430 IF LEN(D$)>=F3 THEN GOTO
7800
7440 GOTO 7200

7600 REM **** BACKSPACE
7610 IF ASC(D1$)<>8 THEN GOTO
7700
7620 COLOR C0,C1,C
7630 IF LEN(D$)=F4 THEN LOCATE
24,1:PRINT P3$;
7640 IF LEN(D$)=0 THEN GOTO 7200
7650 COLOR C4,C5,C
7655 REM --- NEXT LINE, F2$ WON'T
WORK WITH LONG MASK, NEED
MASK VARIABLE
7660 LOCATE
F2,F1+LEN(D$)-1:PRINT F2$;
7670 D$=LEFT$(D$,LEN(D$)-1)
7680 LOCATE F2,F1+LEN(D$)-1
7690 GOTO 7200

7700 REM **** ADD CHR TO STR &
DISPLAY
7710 IF LEN(D$)=F4 THEN GOTO
7200
7720 COLOR C8,C9,C
7730 LOCATE F2,F1+LEN(D$):PRINT
D1$;
7740 D$=D$+D1$

7750 REM **** LENGTH CHECK
7760 IF LEN(D$)<F4+1 THEN GOTO
7200

7800 REM **** RETURN
7810 COLOR C10,C11,C
7820 LOCATE F2,F1:PRINT D$;
7830 COLOR C0,C1,C
7840 PRINT
LEFT$(P3$,F4-LEN(D$)+1);
7850 IF LEN(D$)=F4 THEN LOCATE
24,1:PRINT P3$;
7860 RETURN

7900 REM **** SET CURRENT
CURSOR COLOR SUBROUTINE
(TOGGLE - BLINK)
7910
P2$=CHR$(SCREEN(F2,F1+LEN(D$),0
))
7920
P0=SCREEN(F2,F1+LEN(D$),1):REM
- READ CURRENT COLOR
7930 P1=P0 MOD 16:REM - GET
FOREGROUND VALUE
7940 IF P0>127 THEN LET
P1=P1+16:REM - ADJUST IF
BLINKING
7950 IF P1 = C6 THEN COLOR
C4,C5,C
7960 IF P1 = C4 THEN COLOR
C6,C7,C

```

```

50040 LOCATE 9,1:PRINT "
HEIGHT DISTANCE DISTANCE
dB LOSS TO THROW"
50050 LOCATE 10,1:PRINT "SPEAKER
TO: Feet In. Feet In. Feet In.
decibels degrees"
50060 LOCATE 11,1:PRINT "-----
-----"
50070 LOCATE 18,1:PRINT "-----
-----"

50100 REM *** DISPLAY DATA
50110 REM ** LONG LINES
50120 REM * SET MASK
50130 P1$=":##### ## #####"
## ##### ## #####
###.##"
50140 X=1

50200 REM * DATA
50210 Y=12:P$="LAST LISTENER
":P0#=H1#:P1#=D1#:P2#=T1#:P3#
=L1#:P4#=A1#:GOSUB 51000
50220 Y=13:P$="-6dB LISTENER
":P0#=H2#:P1#=D2#:P2#=T2#:P3#
=L2#:P4#=A2#:GOSUB 51000
50230 Y=14:P$="-12dB
LISTENER":P0#=H3#:P1#=D3#:P2#=
T3#:P3#=L3#:P4#=A3#:GOSUB
51000
50240 Y=16:P$="-6dB BACK
WALL":P0#=H4#:P1#=D5#:P2#=T4#
:P3#=L4#:P4#=A7#:GOSUB 51000
50250 Y=17:P$="-12dB BK WALL
":P0#=H5#:P1#=D5#:P2#=T5#:P3#
=L5#:P4#=A8#:GOSUB 51000

50300 REM ** SINGLE DATA LINES
50310 REM * BOTTOM OF SCREEN
50310 X=1:Y=19:P$="-6dB
COVERAGE ANGLE : #####.##
deg.":P0#=A4#:GOSUB 51200
50320 X=1:Y=20:P$="-12dB
COVERAGE ANGLE : #####.##
deg.":P0#=A5#:GOSUB 51200

50330 X=1:Y=21:P$="SPEAKER TILT
FROM VERT: #####.##
deg.":P0#=A6#:GOSUB 51200
50340 D#=H7#:GOSUB 51500
50350 X=38:Y=19:P$="VERT FLR TO
CENT OF SPKR: #####
ft.":P0#=DF#:GOSUB 51200
50360 X=75:Y=19:P$="##
in.":P0#=DI#:GOSUB 51200
50370 D#=H1#:GOSUB 51500
50380 X=38:Y=20:P$="AVG LIST TO
CENT OF SPKR: #####
ft.":P0#=DF#:GOSUB 51200
50390 X=75:Y=20:P$="##
in.":P0#=DI#:GOSUB 51200

50400 REM * REFRESH TOP OF
SCREEN (INPUTS)
50410 REM HORIZ DIST FROM FIRST
TO LAST LISTENER
50420 D#=D6#:GOSUB 51500
50430
X=22:Y=4:P$="#####":P0#=DF#:GO
SUB 51200
50440 X=30:Y=4:P$="
##":P0#=DI#:GOSUB 51200
50450 REM HORIZ DIST FROM LAST
LIST TO BACK WALL
50460 D#=D4#:GOSUB 51500
50470
X=22:Y=5:P$="#####":P0#=DF#:GO
SUB 51200
50480 X=30:Y=5:P$="
##":P0#=DI#:GOSUB 51200
50500 REM VERT DIST FROM FLOOR
TO AVG LISTENING HEIGHT
50510 D#=H6#:GOSUB 51500
50520
X=63:Y=4:P$="#####":P0#=DF#:GO
SUB 51200
50530 X=71:Y=4:P$="
##":P0#=DI#:GOSUB 51200
50540 REM SPEAKER'S RATED -6dB
COVERAGE ANGLE

50550
X=63:Y=5:P$="#####":P0#=A4#:G
OSUB 51200
50600 REM REFERENCE DISTANCE
50600 D#=R#:GOSUB 51500
50610
X=63:Y=6:P$="#####":P0#=DF#:GO
SUB 51200
50620 X=71:Y=6:P$="
##":P0#=DI#:GOSUB 51200

50999 RETURN

51000 REM **** PRINT LONG LINE
SUBROUTINE
51010 REM *** SET VARIABLES
51020 D#=P0#:GOSUB
51500:PF0#=DF#:PI0#=DI#
51030 D#=P1#:GOSUB
51500:PF1#=DF#:PI1#=DI#
51040 D#=P2#:GOSUB
51500:PF2#=DF#:PI2#=DI#
51110 REM *** PRINT
51110 LOCATE Y,X:PRINT USING
P$+P1$;PF0#,PI0#,PF1#,PI1#,PF2#,P
I2#,P3#,P4#
51120 RETURN

51200 REM **** PRINT A SINGLE LINE
51210 LOCATE Y,X:PRINT USING
P$;P0#
51220 RETURN

51500 REM **** CONVERT TO FEET
AND INCHES
51510 DF#=INT(D#/12)
51520 DI#=INT(D#-(DF#*12)+0.5)
51530 IF DI# < 12 THEN RETURN
51540 DF#=DF#+1
51550 DI#=0
51560 RETURN

65535 END

```